
/********************************************************************** 
SAS Exercises 
Exercise 1. Read NAMCS data & create a SAS dataset   
**********************************************************************/ 
 
filename nam06 'c:\duc\namcs06';                  /*NAMCS06 public use data file*/ 
filename nam06inp 'c:\duc\nam06inp.txt';          /*Input statement to read data file*/ 
filename nam06lab 'c:\duc\nam06lab.txt';          /*Assigns variable labels and formats*/ 
filename nam06for 'c:\duc\nam06for.txt';          /*Value labels*/ 
 
libname out1 'c:\myfiles';                        /*Destination for saved output*/ 
 
%inc nam06for; 
data nam1; 
infile nam06 missover lrecl=999; 
%inc nam06inp; 
%inc nam06lab; 
patwt1k=patwt/1000; 
keep ager sex med1-med8 drugid1--rx8v3c4 patwt1k cstratm cpsum patwt phycode patcode; 
run; 
 
proc print data=nam1 (obs=25); 
title 'Exercise 1. 2006 NAMCS Public Use File - Selected Variables'; 
var phycode patcode med1--med8 drugid1--rx8v3c4 patwt1k; 
run; 
 



/********************************************************************** 
Exercise 2 - Generating a ranked listing of MED codes (drugs as written) 
with 2006 NAMCS public use data  -- demonstrates the 'exploded' file technique -- 
outputs one record for each drug mention 
**********************************************************************/ 
 
data nam2; set nam1; 
  array medcod(8) med1-med8; 
    medcode='     '; 
    do i=1 to 8; 
     if medcod(i) ne '90000' then do; 
           medcode=medcod(i); 
               output; 
        end; 
      end; 
run; 
 
/*compare difference with printout from Exercise 1*/ 
proc print data=nam2 (obs=25);    
 
 
title 'Exercise 2a. 2006 NAMCS Public Use File - Selected Variables'; 
var phycode patcode medcode med1-med8 drugid1-drugid8; 
format med1-med8 drugid1-drugid8; 
run; 
 
 
/****************************************************************************/ 
/*provides a ranked listing, but there are caveats for use of medcode*/ 
proc freq data=nam2 order=freq; 
tables medcode/list missing; 
format medcode $medcodf.; 
weight patwt1k; 
Title "Exercise 2b. 2006 NAMCS ranked drugs by entry name"; 
run; 
 



/*********************************************************************** 
Exercise 3 - Generating a ranked listing of DRUGID (generic) codes 
with 2006 NAMCS public use data -- demonstrates the 'exploded' file 
technique -- outputs one record for each drugid 
 
DRUGID1-DRUGID8 is a much better variable to use than MED1-MED8 
for general research because it is more comprehensive.   
 
You can carry this logic out to the rest of the drug characteristics 
as well, to get not just a file where each record is a drugid, but 
each drugid also has its associated characteristics.  Good for doing 
summary analysis. 
**********************************************************************/ 
 
data nam3; set nam1; 
  array gencod(8) drugid1-drugid8; 
    drugid='      '; 
    do i=1 to 8; 
     if gencod(i) > ' ' then do; 
           drugid=gencod(i); 
               output; 
        end; 
      end; 
run; 
 
data nam4; set nam3;  /*necessary for truncating those long drugid labels*/ 
gentext=substr(put(drugid,$drugidf.),1,30); 
run; 
 
proc freq data=nam4 order=freq; 
tables gentext/list missing; 
weight patwt1k; 
Title "Exercise 3. 2006 NAMCS ranked generic substances (labels truncated to 30 characters)"; 
run; 
 



/********************************************************************** 
Exercise 4. The relationship between RX_CAT_ and Level 1, 2, and 3 
variables - uses visit file, not exploded drug file 
**********************************************************************/ 
 
proc print data=nam1 (obs=35); 
title 'Exercise 4. 2006 NAMCS Public Use File - Showing Therapeutic Levels'; 
var med1 drugid1 rx1cat1 rx1v1c1 rx1v2c1 rx1v3c1; 
format med1 drugid1 med1--med8 rx1cat1--rx1cat4 rx1v1c1--rx1v3c4;  
run; 



/********************************************************************** 
Exercise 5. Creates concatenated therapeutic category variables that 
show the entire nest structure of the Multum classification, using visit file 
**********************************************************************/ 
 
data namcat; set nam1; 
rx1tot1=rx1v1c1||rx1v2c1||rx1v3c1; 
rx1tot2=rx1v1c2||rx1v2c2||rx1v3c2; 
rx1tot3=rx1v1c3||rx1v2c3||rx1v3c3; 
rx1tot4=rx1v1c4||rx1v2c4||rx1v3c4; 
 
rx2tot1=rx2v1c1||rx2v2c1||rx2v3c1; 
rx2tot2=rx2v1c2||rx2v2c2||rx2v3c2; 
rx2tot3=rx2v1c3||rx2v2c3||rx2v3c3; 
rx2tot4=rx2v1c4||rx2v2c4||rx2v3c4; 
 
rx3tot1=rx3v1c1||rx3v2c1||rx3v3c1; 
rx3tot2=rx3v1c2||rx3v2c2||rx3v3c2; 
rx3tot3=rx3v1c3||rx3v2c3||rx3v3c3; 
rx3tot4=rx3v1c4||rx3v2c4||rx3v3c4; 
 
rx4tot1=rx4v1c1||rx4v2c1||rx4v3c1; 
rx4tot2=rx4v1c2||rx4v2c2||rx4v3c2; 
rx4tot3=rx4v1c3||rx4v2c3||rx4v3c3; 
rx4tot4=rx4v1c4||rx4v2c4||rx4v3c4; 
 
rx5tot1=rx5v1c1||rx5v2c1||rx5v3c1; 
rx5tot2=rx5v1c2||rx5v2c2||rx5v3c2; 
rx5tot3=rx5v1c3||rx5v2c3||rx5v3c3; 
rx5tot4=rx5v1c4||rx5v2c4||rx5v3c4; 
 
rx6tot1=rx6v1c1||rx6v2c1||rx6v3c1; 
rx6tot2=rx6v1c2||rx6v2c2||rx6v3c2; 
rx6tot3=rx6v1c3||rx6v2c3||rx6v3c3; 
rx6tot4=rx6v1c4||rx6v2c4||rx6v3c4; 
 
rx7tot1=rx7v1c1||rx7v2c1||rx7v3c1; 
rx7tot2=rx7v1c2||rx7v2c2||rx7v3c2; 



rx7tot3=rx7v1c3||rx7v2c3||rx7v3c3; 
rx7tot4=rx7v1c4||rx7v2c4||rx7v3c4; 
 
rx8tot1=rx8v1c1||rx8v2c1||rx8v3c1; 
rx8tot2=rx8v1c2||rx8v2c2||rx8v3c2; 
rx8tot3=rx8v1c3||rx8v2c3||rx8v3c3; 
rx8tot4=rx8v1c4||rx8v2c4||rx8v3c4; 
 
data namcat2; set namcat; 
 
array rxallcat(32)rx1tot1 rx1tot2 rx1tot3 rx1tot4 
                  rx2tot1 rx2tot2 rx2tot3 rx2tot4 
                                  rx3tot1 rx3tot2 rx3tot3 rx3tot4 
                                  rx4tot1 rx4tot2 rx4tot3 rx4tot4 
                                  rx5tot1 rx5tot2 rx5tot3 rx5tot4 
                                  rx6tot1 rx6tot2 rx6tot3 rx6tot4 
                                  rx7tot1 rx7tot2 rx7tot3 rx7tot4 
                                  rx8tot1 rx8tot2 rx8tot3 rx8tot4; 
 
    therall='000000000'; 
    do i=1 to 32; 
     if rxallcat(i) > '000000000' then do; 
           therall=rxallcat(i); 
               output; 
    end; end; 
run; 
 
proc freq data=namcat2; 
tables therall/list missing; 
title 'Exercise 5. Concatenated therapeutic levels to see entire nest structure'; 
weight patwt1k; 
run; 
 



/********************************************************************* 
Exercise 6.1 Conceptualizing drug data -- Visits vs. Drug Mentions 
At how many visits were antidepressants reported? 
**********************************************************************/ 
data nam6; set nam1; 
if rx1v2c1='249' or    /*selects visits where antidepressants are reported*/ 
rx1v2c2='249' or       /*illustrates a use of 2nd-level therapeutic drug category*/ 
rx1v2c3='249' or 
rx1v2c4='249' or 
 
rx2v2c1='249' or   
rx2v2c2='249' or 
rx2v2c3='249' or 
rx2v2c4='249' or 
 
rx3v2c1='249' or   
rx3v2c2='249' or 
rx3v2c3='249' or 
rx3v2c4='249' or 
rx4v2c1='249' or   
rx4v2c2='249' or 
rx4v2c3='249' or 
rx4v2c4='249' or 
rx5v2c1='249' or   
rx5v2c2='249' or 
rx5v2c3='249' or 
rx5v2c4='249' or 
rx6v2c1='249' or   
rx6v2c2='249' or 
rx6v2c3='249' or 
rx6v2c4='249' or 
rx7v2c1='249' or   
rx7v2c2='249' or 
rx7v2c3='249' or 
rx7v2c4='249' or 
rx8v2c1='249' or   
rx8v2c2='249' or 
rx8v2c3='249' or 



rx8v2c4='249' then antidep=1; 
else antidep=0; 
 
proc surveyfreq data=nam6; 
cluster cpsum; 
strata cstratm; 
weight patwt1k; 
tables antidep*sex; 
Title "Exercise 6.1. 2006 NAMCS Visits with Antidepressant Drug Mentions by Sex"; 
run; 
 
 
 



/********************************************************************** 
Exercise 6.2a Estimate total number of fluoxetine occurrences. 
Will include any drugs reported on the Patient Record form 
that have a DRUGID of fluoxetine in the Multum Lexicon database. 
(Could include entries on the form of 'Prozac', 'fluoxetine', 
'Sarafem' or any other entry that is generically equivalent to 
fluoxetine.) 
**********************************************************************/ 
 
data nam6b; set nam1; 
total=0; 
array gen(8) drugid1-drugid8; 
do i=1 to 8; 
if gen(i)= 'd00236' then total=total+1; 
end; 
proc surveymeans data=nam6b nobs sum mean; 
cluster cpsum; 
strata cstratm; 
var total; 
weight patwt1k; 
Title "Exercise 6.2a. 2006 NAMCS Fluoxetine Occurrences"; 
run; 
 
 



/********************************************************************** 
Exercise 6.2b. Estimate total number of Prozac mentions 
**********************************************************************/ 
 
data nam6c; set nam1; 
total=0; 
array meds(8) med1-med8; 
do i=1 to 8; 
if meds(i)= '25674' then total=total+1; 
end; 
proc surveymeans data=nam6c nobs sum mean; 
cluster cpsum; 
strata cstratm; 
var total; 
weight patwt1k; 
Title "Exercise 6.2b. 2006 NAMCS Prozac Mentions"; 
run; 
 



/********************************************************************* 
Exercise 6.3. Estimate total number of occurrences of the 19 Multum 
Level 1 therapeutic categories -- similar to NDC 2-digit categories 
**********************************************************************/ 
 
data nam6c; set nam1; 
  array meds(8) med1-med8; 
  array rxcat(8,4) 
        rx1v1c1 rx1v1c2 rx1v1c3 rx1v1c4 
        rx2v1c1 rx2v1c2 rx2v1c3 rx2v1c4 
        rx3v1c1 rx3v1c2 rx3v1c3 rx3v1c4 
        rx4v1c1 rx4v1c2 rx4v1c3 rx4v1c4 
        rx5v1c1 rx5v1c2 rx5v1c3 rx5v1c4 
        rx6v1c1 rx6v1c2 rx6v1c3 rx6v1c4 
        rx7v1c1 rx7v1c2 rx7v1c3 rx7v1c4 
        rx8v1c1 rx8v1c2 rx8v1c3 rx8v1c4 
; 
    rxcatx=0; 
    do i=1 to 8; 
                do j=1 to 4; 
                if rxcat(i,j) > '  ' then do; 
                        rxcatx=rxcat(i,j); 
                        mednum=i; 
                        mclcode=meds(i); 
                        output; 
                end; 
                end; 
        end; 
run; 
 
data nam6d; set nam6c; 
serialx=cats(phycode,'-',patcode); 
run; 
 
proc sort data=nam6d nodupkey; 
by serialx mclcode mednum rxcatx; 
run; 
 



proc summary data=nam6d; 
var patwt; 
class rxcatx; 
output out=summed sum=; 
data summed; 
set summed; 
retain denom; 
if _N_ = 1 then denom=patwt; 
perc=(patwt/denom)*100; 
rxcatxx=put(rxcatx,Z3.); 
proc print noobs; 
var rxcatxx _freq_ patwt perc; 
title 'Exercise 6.3 Therapeutic categories - First Level'; 
format rxcatxx $DRGLV1F. patwt comma13. perc 7.3; run; 
 



/********************************************************************** 
Exercise 6.4. Estimate total number of occurrences of 2nd- and 
3rd-level therapeutic categories for antidepressants 
**********************************************************************/ 
 
data level2 level3; set nam1; 
array meds(8) med1-med8; 
array rxcatl2(8,4) 
rx1v2c1 rx1v2c2 rx1v2c3 rx1v2c4 
rx2v2c1 rx2v2c2 rx2v2c3 rx2v2c4 
rx3v2c1 rx3v2c2 rx3v2c3 rx3v2c4 
rx4v2c1 rx4v2c2 rx4v2c3 rx4v2c4 
rx5v2c1 rx5v2c2 rx5v2c3 rx5v2c4 
rx6v2c1 rx6v2c2 rx6v2c3 rx6v2c4 
rx7v2c1 rx7v2c2 rx7v2c3 rx7v2c4 
rx8v2c1 rx8v2c2 rx8v2c3 rx8v2c4 
; 
 
array rxcatl3(8,4) 
rx1v3c1 rx1v3c2 rx1v3c3 rx1v3c4 
rx2v3c1 rx2v3c2 rx2v3c3 rx2v3c4 
rx3v3c1 rx3v3c2 rx3v3c3 rx3v3c4 
rx4v3c1 rx4v3c2 rx4v3c3 rx4v3c4 
rx5v3c1 rx5v3c2 rx5v3c3 rx5v3c4 
rx6v3c1 rx6v3c2 rx6v3c3 rx6v3c4 
rx7v3c1 rx7v3c2 rx7v3c3 rx7v3c4 
rx8v3c1 rx8v3c2 rx8v3c3 rx8v3c4 
; 
 
/* looking for level 2 antidepressents */ 
rxcatxl2=0; 
do i=1 to 8; 
do j=1 to 4; 
if rxcatl2(i,j) =('249') then do; 
rxcatxl2=rxcatl2(i,j); 
mednum=i; 
mclcode=meds(i); 
serialx=cats(phycode,'-',patcode); 



output level2; 
end; 
end; 
end; 
 
/* looking for level 3 antidepressents */ 
i=0; 
j=0; 
rxcatxl3=0; 
do i=1 to 8; 
do j=1 to 4; 
if rxcatl3(i,j) in ('076','208', '209','250','306','307','308') then do; 
rxcatxl3=rxcatl3(i,j); 
mednum=i; 
mclcode=meds(i); 
serialx=cats(phycode,'-',patcode); 
output level3; 
end; 
end; 
end; 
run; 
 
proc sort data=level2 nodupkey; 
by serialx mclcode mednum rxcatxl2; 
run; 
 
proc summary data=level2; 
var patwt; 
class rxcatxl2; 
output out=suml2 sum=; 
data suml2; 
set suml2; 
retain denom; 
if _N_ = 1 then denom=patwt; 
perc=(patwt/denom)*100; 
rxcatxx=left(put(rxcatxl2,Z3.)); 
title 'Exercise 6.4 Number of occurrences of antidepressants - Level 2'; 
 



proc print noobs; 
var rxcatxx _freq_ patwt perc; 
format rxcatxx $DRGLV2F. patwt comma13. perc 7.3; run; 
 
proc sort data=level3 nodupkey; 
by serialx mclcode mednum rxcatxl3; 
run; 
 
proc summary data=level3; 
var patwt; 
class rxcatxl3; 
output out=suml3 sum=; 
 
data suml3; 
set suml3; 
retain denom; 
if _N_ = 1 then denom=patwt; 
perc=(patwt/denom)*100; 
rxcatxx=left(put(rxcatxl3,Z3.)); 
 
proc print noobs; 
var rxcatxx _freq_ patwt perc; 
title 'Exercise 6.4 Number of occurrences of antidepressant subcategories - Level 3'; 
format rxcatxx $DRGLV3F. patwt comma13. perc 7.3; run; 
 
 
data both; 
length rxcatxx $75; 
set suml2 suml3; 
if rxcatxl2 ne . then rxcatxx=left(put(rxcatxx,$DRGLV2F.)); 
if rxcatxl3 ne . then rxcatxx=left(put(rxcatxx,$DRGLV3F.)); 
if rxcatxx = '.' then delete; 
 
proc print data=both noobs; 
var rxcatxx _freq_ patwt perc; 
format patwt comma13. perc 7.3; 
title 'Exercise 6.4 Number of occurrences of antidepressant subcategories - Levels 2 and 3'; 
run; 



/***************************************************************** 
Exercise 7.1 Adding ingredients to the public use file 
******************************************************************/ 
 
 
OPTIONS MERGENOBY=ERROR MSGLEVEL=I; 
 
/*  NOTE: THESE PATHS WILL NEED TO CHANGE DEPENDING ON WHERE DATA AND STATEMENTS ARE 
    STORED ON COMPUTER OR NETWORK. PROGRAM IS CURRENTLY SET UP TO APPLY INGREDIENTS TO  
    THE 2006 NAMCS PUBLIC USE FILE.  CHANGE INPUT DATA FILE AND INPUT FILE STATEMENT BELOW  
    FOR USE WITH THE 2006 NHAMCS PUBLIC USE FILES (EMERGENCY DEPARTMENT AND OUTPATIENT  
    DEPARTMENT FILES).  THIS PROGRAM ASSUMES YOU ARE USING THE SAS INPUT STATEMENTS  
    AVAILABLE AT THE AMBULATORY HEALTH CARE WEB SITE FOR NAMCS AND NHAMCS.  
    (www.cdc.gov/nchs/namcs or www.cdc.gov/nchs/nhamcs) 
    PLEASE CONTACT THE AMBULATORY AND HOSPITAL CARE STATISTICS BRANCH AT 
    301-458-4600 IF YOU HAVE QUESTIONS. 
*/ 
 
FILENAME DRGINGR       'c:\duc\drug_ingredient.data';       *DRUG INGREDIENT FILE; 
FILENAME VIS       'c:\duc\namcs06';                   *INPUT DATA FILE; 
FILENAME INPUT      'c:\duc\nam06inp.txt';               *INPUT STATEMENT FOR INPUT DATA FILE; 
 
LIBNAME  LIBOUT      'c:\duc';             *PATH FOR OUTPUT DATASET; 
 
%LET VISOUT=NEWFILE;  
 
PROC DATASETS LIBRARY=WORK KILL; 
QUIT; 
 
/* CREATING VISIT FILE */ 
DATA VIS; 
INFILE VIS LRECL=999 MISSOVER; 
%INC INPUT; 
RUN; 
 
/* CREATING DRUG INGREDIENT FILE */ 
 
data drging; 



INFILE DRGINGR LRECL=999 MISSOVER PAD; 
INPUT 
@001 DRUG_ID $CHAR6. 
@007 MEMBER1  $CHAR6. 
@013 MEMBER2  $CHAR6. 
@019 MEMBER3  $CHAR6. 
@025 MEMBER4  $CHAR6. 
@031 MEMBER5  $CHAR6. 
@037 MEMBER6  $CHAR6. 
@043 MEM1CAT1  $CHAR3. 
@046 MEM1CAT2 $CHAR3. 
@049 MEM1CAT3 $CHAR3. 
@052 MEM1CAT4 $CHAR3. 
@055 MEM2CAT1 $CHAR3. 
@058 MEM2CAT2 $CHAR3. 
@061 MEM2CAT3 $CHAR3. 
@064 MEM2CAT4 $CHAR3. 
@067 MEM3CAT1 $CHAR3. 
@070 MEM3CAT2 $CHAR3. 
@073 MEM3CAT3 $CHAR3. 
@076 MEM3CAT4 $CHAR3. 
@079 MEM4CAT1  $CHAR3. 
@082 MEM4CAT2 $CHAR3. 
@085 MEM4CAT3 $CHAR3. 
@088 MEM4CAT4 $CHAR3. 
@091 MEM5CAT1 $CHAR3. 
@094 MEM5CAT2 $CHAR3. 
@097 MEM5CAT3 $CHAR3. 
@100 MEM5CAT4 $CHAR3. 
@103 MEM6CAT1 $CHAR3. 
@106 MEM6CAT2  $CHAR3. 
@109 MEM6CAT3 $CHAR3. 
@112 MEM6CAT4 $CHAR3. 
; 
run; 
 
data drging; 
set drging; 



length iddata $36 mem1 mem2 mem3 mem4 mem5 mem6 $12; 
iddata=member1||member2||member3||member4||member5||member6; 
memall= mem1cat1||mem1cat2||mem1cat3||mem1cat4|| 
  mem2cat1||mem2cat2||mem2cat3||mem2cat4|| 
  mem3cat1||mem3cat2||mem3cat3||mem3cat4|| 
  mem4cat1||mem4cat2||mem4cat3||mem4cat4|| 
  mem5cat1||mem5cat2||mem5cat3||mem5cat4|| 
  mem6cat1||mem6cat2||mem6cat3||mem6cat4; 
run; 
 
/* creating temporary format for member drug ids */ 
data memform(keep=start label fmtname hlo); 
set drging end=eof; 
fmtname = '$IDDATA'; 
start = drug_id; 
label = iddata; 
output; 
if eof then do; 
 label = 'NOMATCH'; 
 hlo = 'O'; 
 output; 
 end; 
 
data memall(keep=start label fmtname); 
set drging; 
length label $72; 
fmtname = '$MEMALL'; 
start = drug_id; 
label = memall; 
 
PROC FORMAT CNTLIN=memform LIBRARY=WORK; 
run; 
 
proc format cntlin=memall library=work; 
run; 
 
/* MERGING INGREDIENTS TO VISIT FILE */ 
 



%macro adddata; 
data visingr; 
set vis; 
length memberdata $36 catdata $72;  
array meds(8) med1-med8; 
array drgid(8) drugid1-drugid8; 
array memberid(8,6)$6  drg1mem1-drg1mem6 
      drg2mem1-drg2mem6 
      drg3mem1-drg3mem6 
      drg4mem1-drg4mem6 
      drg5mem1-drg5mem6 
      drg6mem1-drg6mem6 
      drg7mem1-drg7mem6 
      drg8mem1-drg8mem6; 
 
array drg1(24) $3  drg1mem1cat1 drg1mem1cat2 drg1mem1cat3 drg1mem1cat4 
     drg1mem2cat1 drg1mem2cat2 drg1mem2cat3 drg1mem2cat4 
     drg1mem3cat1 drg1mem3cat2 drg1mem3cat3 drg1mem3cat4 
     drg1mem4cat1 drg1mem4cat2 drg1mem4cat3 drg1mem4cat4 
     drg1mem5cat1 drg1mem5cat2 drg1mem5cat3 drg1mem5cat4 
     drg1mem6cat1 drg1mem6cat2 drg1mem6cat3 drg1mem6cat4; 
 
array drg2(24) $3  drg2mem1cat1 drg2mem1cat2 drg2mem1cat3 drg2mem1cat4 
     drg2mem2cat1 drg2mem2cat2 drg2mem2cat3 drg2mem2cat4 
     drg2mem3cat1 drg2mem3cat2 drg2mem3cat3 drg2mem3cat4 
     drg2mem4cat1 drg2mem4cat2 drg2mem4cat3 drg2mem4cat4 
     drg2mem5cat1 drg2mem5cat2 drg2mem5cat3 drg2mem5cat4 
     drg2mem6cat1 drg2mem6cat2 drg2mem6cat3 drg2mem6cat4; 
 
array drg3(24) $3  drg3mem1cat1 drg3mem1cat2 drg3mem1cat3 drg3mem1cat4 
     drg3mem2cat1 drg3mem2cat2 drg3mem2cat3 drg3mem2cat4 
     drg3mem3cat1 drg3mem3cat2 drg3mem3cat3 drg3mem3cat4 
     drg3mem4cat1 drg3mem4cat2 drg3mem4cat3 drg3mem4cat4 
     drg3mem5cat1 drg3mem5cat2 drg3mem5cat3 drg3mem5cat4 
     drg3mem6cat1 drg3mem6cat2 drg3mem6cat3 drg3mem6cat4; 
 
array drg4(24) $3  drg4mem1cat1 drg4mem1cat2 drg4mem1cat3 drg4mem1cat4 
     drg4mem2cat1 drg4mem2cat2 drg4mem2cat3 drg4mem2cat4 



     drg4mem3cat1 drg4mem3cat2 drg4mem3cat3 drg4mem3cat4 
     drg4mem4cat1 drg4mem4cat2 drg4mem4cat3 drg4mem4cat4 
     drg4mem5cat1 drg4mem5cat2 drg4mem5cat3 drg4mem5cat4 
     drg4mem6cat1 drg4mem6cat2 drg4mem6cat3 drg4mem6cat4; 
 
array drg5(24) $3  drg5mem1cat1 drg5mem1cat2 drg5mem1cat3 drg5mem1cat4 
     drg5mem2cat1 drg5mem2cat2 drg5mem2cat3 drg5mem2cat4 
     drg5mem3cat1 drg5mem3cat2 drg5mem3cat3 drg5mem3cat4 
     drg5mem4cat1 drg5mem4cat2 drg5mem4cat3 drg5mem4cat4 
     drg5mem5cat1 drg5mem5cat2 drg5mem5cat3 drg5mem5cat4 
     drg5mem6cat1 drg5mem6cat2 drg5mem6cat3 drg5mem6cat4; 
 
array drg6(24) $3  drg6mem1cat1 drg6mem1cat2 drg6mem1cat3 drg6mem1cat4 
     drg6mem2cat1 drg6mem2cat2 drg6mem2cat3 drg6mem2cat4 
     drg6mem3cat1 drg6mem3cat2 drg6mem3cat3 drg6mem3cat4 
     drg6mem4cat1 drg6mem4cat2 drg6mem4cat3 drg6mem4cat4 
     drg6mem5cat1 drg6mem5cat2 drg6mem5cat3 drg6mem5cat4 
     drg6mem6cat1 drg6mem6cat2 drg6mem6cat3 drg6mem6cat4; 
 
array drg7(24) $3  drg7mem1cat1 drg7mem1cat2 drg7mem1cat3 drg7mem1cat4 
     drg7mem2cat1 drg7mem2cat2 drg7mem2cat3 drg7mem2cat4 
     drg7mem3cat1 drg7mem3cat2 drg7mem3cat3 drg7mem3cat4 
     drg7mem4cat1 drg7mem4cat2 drg7mem4cat3 drg7mem4cat4 
     drg7mem5cat1 drg7mem5cat2 drg7mem5cat3 drg7mem5cat4 
     drg7mem6cat1 drg7mem6cat2 drg7mem6cat3 drg7mem6cat4; 
 
array drg8(24) $3  drg8mem1cat1 drg8mem1cat2 drg8mem1cat3 drg8mem1cat4 
     drg8mem2cat1 drg8mem2cat2 drg8mem2cat3 drg8mem2cat4 
     drg8mem3cat1 drg8mem3cat2 drg8mem3cat3 drg8mem3cat4 
     drg8mem4cat1 drg8mem4cat2 drg8mem4cat3 drg8mem4cat4 
     drg8mem5cat1 drg8mem5cat2 drg8mem5cat3 drg8mem5cat4 
     drg8mem6cat1 drg8mem6cat2 drg8mem6cat3 drg8mem6cat4; 
%do i=1 %to 8; 
 if meds(&i) not in ('90000' '88888') then do; 
  memberdata = put(drgid(&i), $IDDATA.); 
  catdata = put(drgid(&i),$MEMALL.); 
  if memberdata = 'NOMATCH' then do; 
   put 'NO MATCH FOR DRUG ID CODE,MISSING DRUG ID ' drgid(&i)=; 



  end; 
  do j = 1 to 6; 
   memberid(&i,j)= substr(memberdata,((6*j)-5),6); 
  end; 
  do k = 1 to 24; 
   drg&i(k)=substr(catdata,((3*k)-2),3); 
  end; 
 end; 
%end;  
 
drop  j k memberdata catdata; 
run; 
%mend; 
%adddata; 
 
DATA LIBOUT.&VISOUT; 
SET VISINGR; 
RUN; 
 
proc contents data=libout.newfile; run; 
 



/********************************************************************* 
Exercise 7.2 Conceptualizing ingredient data -- What difference does the 
ingredient file addition make? (aside from an extra 90+ megabtytes?!) 
**********************************************************************/ 
/*We run the original (non-ingredient) public use dataset to obtain 
the number of visits where a decongestant was provided, prescribed, or continued*/ 
 
 
filename nam06 'k:\acsb_data\namcs06\public.use.data'; /*data file*/ 
filename nam06for 'k:\acsb\publicusefiles\06namcs\sas\nam06for.txt' /*value labels, format assignments*/; 
filename nam06inp 'k:\acsb\publicusefiles\06namcs\sas\nam06inp.txt'; /*input*/ 
filename nam06lab 'k:\acsb\publicusefiles\06namcs\sas\nam06lab.txt' /*variable labels*/; 
%inc nam06for; 
 
data nam1; 
infile nam06 missover lrecl=999; 
%inc nam06inp; 
%inc nam06lab; 
patwt1k=patwt/1000; 
if rx1v2c1='127' or    /*selects visits where decongestants are reported*/ 
rx1v2c2='127' or       /*using the non-ingredient file and examining drug therapeutic classes only*/ 
rx1v2c3='127' or       /*illustrates a use of 2nd-level therapeutic drug category*/ 
rx1v2c4='127' or 
rx2v2c1='127' or   
rx2v2c2='127' or 
rx2v2c3='127' or 
rx2v2c4='127' or 
rx3v2c1='127' or   
rx3v2c2='127' or 
rx3v2c3='127' or 
rx3v2c4='127' or 
rx4v2c1='127' or   
rx4v2c2='127' or 
rx4v2c3='127' or 
rx4v2c4='127' or 
rx5v2c1='127' or   
rx5v2c2='127' or 
rx5v2c3='127' or 



rx5v2c4='127' or 
rx6v2c1='127' or   
rx6v2c2='127' or 
rx6v2c3='127' or 
rx6v2c4='127' or 
rx7v2c1='127' or   
rx7v2c2='127' or 
rx7v2c3='127' or 
rx7v2c4='127' or 
rx8v2c1='127' or   
rx8v2c2='127' or 
rx8v2c3='127' or 
rx8v2c4='127' then deconges=1; 
else deconges=0; 
patwt1k=patwt/1000; 
 
proc surveyfreq data=nam1; 
cluster cpsum; 
strata cstratm; 
weight patwt1k; 
tables deconges*sex; 
title 'Exercise 7.2 Number of visits with decongestant, based only on DRUGID`s therapeutic categories'; 
run; 
 
 



/********************************************************************* 
Exercise 7.3  
We use the new file (public use + ingredients) to look at visits with 
a decongestant reported. This will "tease out" the decongestants from all of the 
combination products that Multum assigned therapeutic classes of  
"respiratory combinations." 
**********************************************************************/ 
 
data ingtest; set libout.newfile; 
if drg1mem1cat1='127' or    /*selects visits where first drug has decongestant ingredients*/ 
drg1mem1cat2='127' or  
drg1mem1cat3='127' or  
drg1mem1cat4='127' or  
 
drg1mem2cat1='127' or  
drg1mem2cat2='127' or  
drg1mem2cat3='127' or  
drg1mem2cat4='127' or  
 
drg1mem3cat1='127' or  
drg1mem3cat2='127' or  
drg1mem3cat3='127' or  
drg1mem3cat4='127' or  
 
drg1mem4cat1='127' or  
drg1mem4cat2='127' or  
drg1mem4cat3='127' or  
drg1mem4cat4='127' or  
 
drg1mem5cat1='127' or  
drg1mem5cat2='127' or  
drg1mem5cat3='127' or  
drg1mem5cat4='127' or  
 
drg1mem6cat1='127' or  
drg1mem6cat2='127' or  
drg1mem6cat3='127' or  
drg1mem6cat4='127' or  



 
drg2mem1cat1='127' or    /*selects visits where 2nd drug has decongestants*/ 
drg2mem1cat2='127' or  
drg2mem1cat3='127' or  
drg2mem1cat4='127' or  
 
drg2mem2cat1='127' or  
drg2mem2cat2='127' or  
drg2mem2cat3='127' or  
drg2mem2cat4='127' or  
 
drg2mem3cat1='127' or  
drg2mem3cat2='127' or  
drg2mem3cat3='127' or  
drg2mem3cat4='127' or  
 
drg2mem4cat1='127' or  
drg2mem4cat2='127' or  
drg2mem4cat3='127' or  
drg2mem4cat4='127' or  
 
drg2mem5cat1='127' or  
drg2mem5cat2='127' or  
drg2mem5cat3='127' or  
drg2mem5cat4='127' or  
 
drg2mem6cat1='127' or  
drg2mem6cat2='127' or  
drg2mem6cat3='127' or  
drg2mem6cat4='127' or  
 
drg3mem1cat1='127' or    /*selects visits where third drug has decongestants*/ 
drg3mem1cat2='127' or  
drg3mem1cat3='127' or  
drg3mem1cat4='127' or  
 
drg3mem2cat1='127' or  
drg3mem2cat2='127' or  



drg3mem2cat3='127' or  
drg3mem2cat4='127' or  
 
drg3mem3cat1='127' or  
drg3mem3cat2='127' or  
drg3mem3cat3='127' or  
drg3mem3cat4='127' or  
 
drg3mem4cat1='127' or  
drg3mem4cat2='127' or  
drg3mem4cat3='127' or  
drg1mem4cat4='127' or  
 
drg3mem5cat1='127' or  
drg3mem5cat2='127' or  
drg3mem5cat3='127' or  
drg3mem5cat4='127' or  
 
drg3mem6cat1='127' or  
drg3mem6cat2='127' or  
drg3mem6cat3='127' or  
drg3mem6cat4='127' or  
 
drg4mem1cat1='127' or    /*selects visits where fourth drug has decongestants*/ 
drg4mem1cat2='127' or  
drg4mem1cat3='127' or  
drg4mem1cat4='127' or  
 
drg4mem2cat1='127' or  
drg4mem2cat2='127' or  
drg4mem2cat3='127' or  
drg4mem2cat4='127' or  
 
drg4mem3cat1='127' or  
drg4mem3cat2='127' or  
drg4mem3cat3='127' or  
drg4mem3cat4='127' or  
 



drg4mem4cat1='127' or  
drg4mem4cat2='127' or  
drg4mem4cat3='127' or  
drg4mem4cat4='127' or  
 
drg4mem5cat1='127' or  
drg4mem5cat2='127' or  
drg4mem5cat3='127' or  
drg4mem5cat4='127' or  
 
drg4mem6cat1='127' or  
drg4mem6cat2='127' or  
drg4mem6cat3='127' or  
drg4mem6cat4='127' or  
 
drg5mem1cat1='127' or    /*selects visits where fifth drug has decongestants*/ 
drg5mem1cat2='127' or  
drg5mem1cat3='127' or  
drg5mem1cat4='127' or  
 
drg5mem2cat1='127' or  
drg5mem2cat2='127' or  
drg5mem2cat3='127' or  
drg5mem2cat4='127' or  
 
drg5mem3cat1='127' or  
drg5mem3cat2='127' or  
drg5mem3cat3='127' or  
drg5mem3cat4='127' or  
 
drg5mem4cat1='127' or  
drg5mem4cat2='127' or  
drg5mem4cat3='127' or  
drg5mem4cat4='127' or  
 
drg5mem5cat1='127' or  
drg5mem5cat2='127' or  
drg5mem5cat3='127' or  



drg5mem5cat4='127' or  
 
drg5mem6cat1='127' or  
drg5mem6cat2='127' or  
drg5mem6cat3='127' or  
drg5mem6cat4='127' or  
 
drg6mem1cat1='127' or    /*selects visits where sixth drug has decongestants*/ 
drg6mem1cat2='127' or  
drg6mem1cat3='127' or  
drg6mem1cat4='127' or  
 
drg6mem2cat1='127' or  
drg6mem2cat2='127' or  
drg6mem2cat3='127' or  
drg6mem2cat4='127' or  
 
drg6mem3cat1='127' or  
drg6mem3cat2='127' or  
drg6mem3cat3='127' or  
drg6mem3cat4='127' or  
 
drg6mem4cat1='127' or  
drg6mem4cat2='127' or  
drg6mem4cat3='127' or  
drg6mem4cat4='127' or  
 
drg6mem5cat1='127' or  
drg6mem5cat2='127' or  
drg6mem5cat3='127' or  
drg6mem5cat4='127' or  
 
drg6mem6cat1='127' or  
drg6mem6cat2='127' or  
drg6mem6cat3='127' or  
drg6mem6cat4='127' or  
 
drg7mem1cat1='127' or    /*selects visits where 7th drug has decongestants*/ 



drg7mem1cat2='127' or  
drg7mem1cat3='127' or  
drg7mem1cat4='127' or  
 
drg7mem2cat1='127' or  
drg7mem2cat2='127' or  
drg7mem2cat3='127' or  
drg7mem2cat4='127' or  
 
drg7mem3cat1='127' or  
drg7mem3cat2='127' or  
drg7mem3cat3='127' or  
drg7mem3cat4='127' or  
 
drg7mem4cat1='127' or  
drg7mem4cat2='127' or  
drg7mem4cat3='127' or  
drg7mem4cat4='127' or  
 
drg7mem5cat1='127' or  
drg7mem5cat2='127' or  
drg7mem5cat3='127' or  
drg7mem5cat4='127' or  
 
drg7mem6cat1='127' or  
drg7mem6cat2='127' or  
drg7mem6cat3='127' or  
drg7mem6cat4='127' or  
 
drg8mem1cat1='127' or    /*selects visits where 8th drug has decongestants*/ 
drg8mem1cat2='127' or  
drg8mem1cat3='127' or  
drg8mem1cat4='127' or  
 
drg8mem2cat1='127' or  
drg8mem2cat2='127' or  
drg8mem2cat3='127' or  
drg8mem2cat4='127' or  



 
drg8mem3cat1='127' or  
drg8mem3cat2='127' or  
drg8mem3cat3='127' or  
drg8mem3cat4='127' or  
 
drg8mem4cat1='127' or  
drg8mem4cat2='127' or  
drg8mem4cat3='127' or  
drg8mem4cat4='127' or  
 
drg8mem5cat1='127' or  
drg8mem5cat2='127' or  
drg8mem5cat3='127' or  
drg8mem5cat4='127' or  
 
drg8mem6cat1='127' or  
drg8mem6cat2='127' or  
drg8mem6cat3='127' or  
drg8mem6cat4='127'   
 
then deconges2=1; 
else deconges2=0; 
patwt1k=patwt/1000; 
run; 
proc surveyfreq data=ingtest; 
cluster cpsum; 
strata cstratm; 
weight patwt1k; 
tables deconges2*sex; 
Title 'Exercise 7.3. 2006 NAMCS Number of visits with decongestants based  
on each ingredient`s therapeutic categories'; 
run; 
 



/***************************************************************** 
Exercise 8. Assigning Multum drug characteristics to data files 
prior to 2006 for multi-year analysis 
******************************************************************/  
DM LOG 'CLEAR'; 
DM OUT 'CLEAR'; 
OPTIONS MERGENOBY=ERROR MSGLEVEL=I; 
 
/*       
THESE PATHS WILL NEED TO CHANGE DEPENDING ON WHERE DATA AND STATEMENTS ARE STORED ON COMPUTER OR NETWORK. 
PROGRAM IS CURRENTLY SET UP TO APPLY 
CHARACTERISTICS FROM 2006 NAMCS/NHAMCS DRUG DATABASE TO 2005 NAMCS PUBLIC USE FILE. YOU WOULD NEED TO 
CHANGE INPUT DATA FILE AND INPUT FILE STATEMENT BELOW FOR USE WITH OTHER NAMCS AND NHAMCS PUBLIC USE FILES.  
THIS PROGRAM ASSUMES YOU ARE USING THE SAS INPUT STATEMENTS AVAILABLE AT THE AMBULATORY HEALTH CARE WEB 
SITE FOR NAMCS AND NHAMCS.  
www.cdc.gov/nchs/namcs.htm and www.cdc.gov/nchs/nhamcs.htm) 
PLEASE CONTACT THE AMBULATORY AND HOSPITAL CARE STATISTICS BRANCH AT 
301-458-4600 IF YOU HAVE QUESTIONS. 
(LAST MODIFIED 8/18/2008) 
*/ 
 
FILENAME MCLFLAT    'C:\DUC\MEDCODE-DRUGIDMAP.DATA';    *DRUG CHARACTERISTICS FILE; 
FILENAME VIS        'C:\DUC\NAMCS05';                   *data file; 
FILENAME INPUT      'C:\DUC\NAM05inp.txt';              *input; 
LIBNAME  LIBOUT   'C:\DUC\';                     *PATH FOR OUTPUT DATASET; 
 
%LET VISOUT=NEWDRUGCHAR;                       *OUTPUT DATASET NAME; 
 
PROC DATASETS LIBRARY=WORK KILL; 
QUIT; 
 
 
/* CREATING DRUG CHARACTERISTICS FILE */ 
DATA MULTUM; 
INFILE MCLFLAT LRECL=999 MISSOVER PAD; 
INPUT 
@001 MCLCODE  $CHAR5. 
@006 MCLNAME  $CHAR70. 



@076 DRUGID  $CHAR6. 
@082 DRUGNAME $CHAR142. 
@224 PRESCR  $CHAR1. 
@225 CONTSUB  $CHAR1. 
@226 COMSTAT  $CHAR1. 
@227 RXCAT1  $CHAR3. 
@230 RXCAT2  $CHAR3. 
@233 RXCAT3  $CHAR3. 
@236 RXCAT4  $CHAR3. 
@239 RXLV1C1  $CHAR3. 
@242 RXLV1C2  $CHAR3. 
@245 RXLV1C3  $CHAR3. 
@248 RXLV1C4  $CHAR3. 
@251 RXLV2C1  $CHAR3. 
@254 RXLV2C2  $CHAR3. 
@257 RXLV2C3  $CHAR3. 
@260 RXLV2C4  $CHAR3. 
@263 RXLV3C1  $CHAR3. 
@266 RXLV3C2  $CHAR3. 
@269 RXLV3C3  $CHAR3. 
@272 RXLV3C4  $CHAR3. 
; 
RUN; 
 
 
/* CREATING TEMPORARY FORMATS FOR DRUG ID*/ 
DATA DRUGID(KEEP=START LABEL FMTNAME HLO); 
SET MULTUM END=EOF; 
LENGTH START $5 LABEL $6 FMTNAME $8; 
FMTNAME = '$DRUGID'; 
START = MCLCODE; 
LABEL = DRUGID; 
OUTPUT; 
 IF EOF THEN DO; 
 LABEL = 'NOMAT'; 
 HLO = 'O'; 
 OUTPUT; 
 END;  



RUN;                                                     
       
 
/* CREATING TEMPORARY FORMATS FOR PRESCRIPTION STATUS */ 
DATA PRESCR(KEEP=START LABEL FMTNAME); 
SET MULTUM; 
LENGTH START $5 LABEL $5 FMTNAME $8; 
FMTNAME = 'PRESCR'; 
START = MCLCODE; 
LABEL = PRESCR; 
 
/* CREATING TEMPORARY FORMATS FOR CONTROLLED STATUS  */ 
DATA CONTSUB(KEEP=START LABEL FMTNAME); 
SET MULTUM; 
LENGTH START $5 LABEL $5 FMTNAME $8; 
FMTNAME = 'CONTSUB'; 
START = MCLCODE; 
LABEL = CONTSUB; 
 
/* CREATING TEMPORARY FORMATS FOR COMPOSITION STATUS  */ 
DATA COMSTAT(KEEP=START LABEL FMTNAME); 
SET MULTUM; 
LENGTH START $5 LABEL $5 FMTNAME $8; 
FMTNAME = 'COMSTAT'; 
START = MCLCODE; 
LABEL = COMSTAT; 
 
 
/* CREATING TEMPORARY FORMATS FOR MULTUM DRUG CATEGORY */ 
DATA RXCAT(KEEP=START LABEL FMTNAME); 
SET MULTUM; 
LENGTH START $5 ALLCAT LABEL $56 FMTNAME $8; 
FMTNAME = '$RXCAT'; 
START = MCLCODE; 
ALLCAT=CAT(RXCAT1,RXCAT2,RXCAT3,RXCAT4,RXLV1C1,RXLV1C2,RXLV1C3,RXLV1C4, 
RXLV2C1,RXLV2C2,RXLV2C3,RXLV2C4,RXLV3C1,RXLV3C2,RXLV3C3,RXLV3C4); 
LABEL = ALLCAT; 
RUN; 



 
DATA NUMCHR; 
SET PRESCR CONTSUB COMSTAT; 
 
PROC FORMAT CNTLIN=NUMCHR LIBRARY=WORK; 
 
PROC FORMAT CNTLIN=DRUGID LIBRARY=WORK; 
 
PROC FORMAT CNTLIN=RXCAT LIBRARY=WORK; 
RUN; 
 
/* MAKING SURE ALL DATA IS REMOVED FROM TEMP LIBARAY */ 
PROC DATASETS LIBRARY=WORK; 
DELETE DRUGID PRESCR CONTSUB COMSTAT RXCAT;  
QUIT; 
RUN; 
 
/* CREATING VISIT FILE */ 
DATA VIS; 
INFILE VIS LRECL=999 MISSOVER; 
%INC INPUT; 
 
RUN; 
 
/* INITIALIZING VARIABLES FOR APPLYING DRUG CHARACTERISTICS */ 
%LET VAR1=; 
%LET VAR2=; 
%LET VAR3=; 
%LET VAR4=; 
%LET VARSTOP=; 
%LET I=; 
%LET J=; 
%LET K=; 
%LET PASSAR=; 
%LET VISYEAR=; 
/*  
DROPPING DRUG CHARACTERISTICS FROM VISIT FILE 
*/ 



PROC SQL /*NOPRINT*/; 
    SELECT NAME 
    INTO: VAR1-:VAR4 
 FROM DICTIONARY.COLUMNS 
    WHERE ((LIBNAME='WORK' AND MEMNAME = 'VIS') AND 
 NAME IN ('DRG5ING5' 'DRG6ING5' 'DINGA25' 'ADRG2IN5' 'DRUG1CL2' 'DRG8ING5')); 
 QUIT; 
 RUN; 
 
 
%MACRO DROPMED; 
%IF &VAR1 = DRG5ING5 %THEN %DO; 
 %LET J = 5; 
%END; 
%IF &VAR2 = DRG6ING5 %THEN %DO; 
 %LET J = 6; 
%END; 
/* VAR3 USED TWICE, ED FILES WILL NOT HAVE BOTH NAMES 
 FOR ADVERSE DRUGS 
*/ 
%IF &VAR3 = DINGA25 OR &VAR3 = ADRG2IN5 OR &VAR4 = DRG8ING5 %THEN %DO; 
 %LET K = 8; 
 %IF &VAR3 = DINGA25 OR &VAR3 = ADRG2IN5 %THEN %DO; 
  %LET VISYEAR = EDONLY; 
 %END; 
%END; 
%IF &K = 8 %THEN %DO; 
 %LET I = 8; 
%END; 
%ELSE %DO; 
 %LET I = &J; 
%END; 
 
DATA VISNDC; 
SET VIS; 
%IF &VAR3 = DINGA25 %THEN %DO; 
 DROP GEN1--&VAR3; 
%END; 



%ELSE %IF &VAR3 = ADRG2IN5 %THEN %DO; 
 DROP GEN1--&VAR3 DRUG1CL2--ADRG2CL3; 
%END; 
%ELSE %IF &VAR3 = DRUG1CL2 %THEN %DO; 
 DROP GEN1--DRG6ING5 DRUG1CL2--DRUG6CL3; 
%END; 
%ELSE %IF &VAR2 = DRG6ING5 %THEN %DO; 
 DROP GEN1--&VAR2; 
%END; 
%ELSE %IF &VAR4 = DRG8ING5 %THEN %DO; 
 DROP GEN1--&VAR4; 
%END; 
%ELSE %DO; 
 DROP GEN1--DRG5ING5; 
%END; 
RUN; 
%MEND; 
%DROPMED; 
RUN; 
 
 
%MACRO CREATEMED; 
DATA VISNDC; 
SET VISNDC; 
LENGTH  DRUGID1-DRUGID&I    $6 
        PRESCR1-PRESCR&I     3 
        CONTSUB1-CONTSUB&I   3 
        COMSTAT1-COMSTAT&I   3 
   
RX1CAT1 RX1CAT2 RX1CAT3 RX1CAT4 
RX2CAT1 RX2CAT2 RX2CAT3 RX2CAT4 
RX3CAT1 RX3CAT2 RX3CAT3 RX3CAT4 
RX4CAT1 RX4CAT2 RX4CAT3 RX4CAT4 
RX5CAT1 RX5CAT2 RX5CAT3 RX5CAT4  
 
RX1V1C1 RX1V1C2 RX1V1C3 RX1V1C4 
RX2V1C1 RX2V1C2 RX2V1C3 RX2V1C4 
RX3V1C1 RX3V1C2 RX3V1C3 RX3V1C4 



RX4V1C1 RX4V1C2 RX4V1C3 RX4V1C4 
RX5V1C1 RX5V1C2 RX5V1C3 RX5V1C4  
 
RX1V2C1 RX1V2C2 RX1V2C3 RX1V2C4 
RX2V2C1 RX2V2C2 RX2V2C3 RX2V2C4 
RX3V2C1 RX3V2C2 RX3V2C3 RX3V2C4 
RX4V2C1 RX4V2C2 RX4V2C3 RX4V2C4 
RX5V2C1 RX5V2C2 RX5V2C3 RX5V2C4 
 
RX1V3C1 RX1V3C2 RX1V3C3 RX1V3C4 
RX2V3C1 RX2V3C2 RX2V3C3 RX2V3C4 
RX3V3C1 RX3V3C2 RX3V3C3 RX3V3C4 
RX4V3C1 RX4V3C2 RX4V3C3 RX4V3C4 
RX5V3C1 RX5V3C2 RX5V3C3 RX5V3C4  $3 
 
CATEGORY $56; 
%IF &I = 5 %THEN %DO; 
%LET PASSAR=  CATEG(5,16) RX1CAT1 RX1CAT2 RX1CAT3 RX1CAT4 RX1V1C1 RX1V1C2 RX1V1C3 RX1V1C4 RX1V2C1 
RX1V2C2 RX1V2C3 RX1V2C4 RX1V3C1 RX1V3C2 RX1V3C3 RX1V3C4 
          RX2CAT1 RX2CAT2 RX2CAT3 RX2CAT4 RX2V1C1 RX2V1C2 RX2V1C3 RX2V1C4 
RX2V2C1 RX2V2C2 RX2V2C3 RX2V2C4 RX2V3C1 RX2V3C2 RX2V3C3 RX2V3C4 
       RX3CAT1 RX3CAT2 RX3CAT3 RX3CAT4 RX3V1C1 RX3V1C2 RX3V1C3 RX3V1C4 
RX3V2C1 RX3V2C2 RX3V2C3 RX3V2C4 RX3V3C1 RX3V3C2 RX3V3C3 RX3V3C4 
       RX4CAT1 RX4CAT2 RX4CAT3 RX4CAT4 RX4V1C1 RX4V1C2 RX4V1C3 RX4V1C4 
RX4V2C1 RX4V2C2 RX4V2C3 RX4V2C4 RX4V3C1 RX4V3C2 RX4V3C3 RX4V3C4 
       RX5CAT1 RX5CAT2 RX5CAT3 RX5CAT4 RX5V1C1 RX5V1C2 RX5V1C3 RX5V1C4 
RX5V2C1 RX5V2C2 RX5V2C3 RX5V2C4 RX5V3C1 RX5V3C2 RX5V3C3 RX5V3C4; 
%END; 
%IF &I = 6 %THEN %DO; 
  LENGTH  RX6CAT1 RX6CAT2 RX6CAT3 RX6CAT4  
    RX6V1C1 RX6V1C2 RX6V1C3 RX6V1C4  
    RX6V2C1 RX6V2C2 RX6V2C3 RX6V2C4 
    RX6V3C1 RX6V3C2 RX6V3C3 RX6V3C4  $3; 
%LET PASSAR =  CATEG(6,16)  RX1CAT1 RX1CAT2 RX1CAT3 RX1CAT4 RX1V1C1 RX1V1C2 RX1V1C3 RX1V1C4 RX1V2C1 
RX1V2C2 RX1V2C3 RX1V2C4 RX1V3C1 RX1V3C2 RX1V3C3 RX1V3C4 
        RX2CAT1 RX2CAT2 RX2CAT3 RX2CAT4 RX2V1C1 RX2V1C2 RX2V1C3 
RX2V1C4 RX2V2C1 RX2V2C2 RX2V2C3 RX2V2C4 RX2V3C1 RX2V3C2 RX2V3C3 RX2V3C4 



        RX3CAT1 RX3CAT2 RX3CAT3 RX3CAT4 RX3V1C1 RX3V1C2 RX3V1C3 
RX3V1C4 RX3V2C1 RX3V2C2 RX3V2C3 RX3V2C4 RX3V3C1 RX3V3C2 RX3V3C3 RX3V3C4 
        RX4CAT1 RX4CAT2 RX4CAT3 RX4CAT4 RX4V1C1 RX4V1C2 RX4V1C3 
RX4V1C4 RX4V2C1 RX4V2C2 RX4V2C3 RX4V2C4 RX4V3C1 RX4V3C2 RX4V3C3 RX4V3C4 
        RX5CAT1 RX5CAT2 RX5CAT3 RX5CAT4 RX5V1C1 RX5V1C2 RX5V1C3 
RX5V1C4 RX5V2C1 RX5V2C2 RX5V2C3 RX5V2C4 RX5V3C1 RX5V3C2 RX5V3C3 RX5V3C4 
        RX6CAT1 RX6CAT2 RX6CAT3 RX6CAT4 RX6V1C1 RX6V1C2 RX6V1C3 
RX6V1C4 RX6V2C1 RX6V2C2 RX6V2C3 RX6V2C4 RX6V3C1 RX6V3C2 RX6V3C3 RX6V3C4; 
%END; 
%IF &I = 8 %THEN %DO; 
 LENGTH  RX6CAT1 RX6CAT2 RX6CAT3 RX6CAT4 
   RX6V1C1 RX6V1C2 RX6V1C3 RX6V1C4 
   RX6V2C1 RX6V2C2 RX6V2C3 RX6V2C4 
   RX6V3C1 RX6V3C2 RX6V3C3 RX6V3C4 
   RX7CAT1 RX7CAT2 RX7CAT3 RX7CAT4 
   RX7V1C1 RX7V1C2 RX7V1C3 RX7V1C4 
   RX7V2C1 RX7V2C2 RX7V2C3 RX7V2C4 
   RX7V3C1 RX7V3C2 RX7V3C3 RX7V3C4 
   RX8CAT1 RX8CAT2 RX8CAT3 RX8CAT4  
   RX8V1C1 RX8V1C2 RX8V1C3 RX8V1C4 
   RX8V2C1 RX8V2C2 RX8V2C3 RX8V2C4  
   RX8V3C1 RX8V3C2 RX8V3C3 RX8V3C4  $3; 
 
%LET PASSAR =  CATEG(8,16) RX1CAT1 RX1CAT2 RX1CAT3 RX1CAT4 RX1V1C1 RX1V1C2 RX1V1C3 RX1V1C4 RX1V2C1 
RX1V2C2 RX1V2C3 RX1V2C4 RX1V3C1 RX1V3C2 RX1V3C3 RX1V3C4 
        RX2CAT1 RX2CAT2 RX2CAT3 RX2CAT4 RX2V1C1 RX2V1C2 RX2V1C3 
RX2V1C4 RX2V2C1 RX2V2C2 RX2V2C3 RX2V2C4 RX2V3C1 RX2V3C2 RX2V3C3 RX2V3C4 
        RX3CAT1 RX3CAT2 RX3CAT3 RX3CAT4 RX3V1C1 RX3V1C2 RX3V1C3 
RX3V1C4 RX3V2C1 RX3V2C2 RX3V2C3 RX3V2C4 RX3V3C1 RX3V3C2 RX3V3C3 RX3V3C4 
        RX4CAT1 RX4CAT2 RX4CAT3 RX4CAT4 RX4V1C1 RX4V1C2 RX4V1C3 
RX4V1C4 RX4V2C1 RX4V2C2 RX4V2C3 RX4V2C4 RX4V3C1 RX4V3C2 RX4V3C3 RX4V3C4 
        RX5CAT1 RX5CAT2 RX5CAT3 RX5CAT4 RX5V1C1 RX5V1C2 RX5V1C3 
RX5V1C4 RX5V2C1 RX5V2C2 RX5V2C3 RX5V2C4 RX5V3C1 RX5V3C2 RX5V3C3 RX5V3C4 
        RX6CAT1 RX6CAT2 RX6CAT3 RX6CAT4 RX6V1C1 RX6V1C2 RX6V1C3 
RX6V1C4 RX6V2C1 RX6V2C2 RX6V2C3 RX6V2C4 RX6V3C1 RX6V3C2 RX6V3C3 RX6V3C4 
        RX7CAT1 RX7CAT2 RX7CAT3 RX7CAT4 RX7V1C1 RX7V1C2 RX7V1C3 
RX7V1C4 RX7V2C1 RX7V2C2 RX7V2C3 RX7V2C4 RX7V3C1 RX7V3C2 RX7V3C3 RX7V3C4 



        RX8CAT1 RX8CAT2 RX8CAT3 RX8CAT4 RX8V1C1 RX8V1C2 RX8V1C3 
RX8V1C4 RX8V2C1 RX8V2C2 RX8V2C3 RX8V2C4 RX8V3C1 RX8V3C2 RX8V3C3 RX8V3C4; 
%END; 
%IF &VISYEAR = EDONLY %THEN %DO; 
 MED7=ADVDRUG1; 
 MED8=ADVDRUG2; 
%END; 
 
ARRAY MEDS(&I)      MED1-MED&I; 
ARRAY DRGID(&I)     DRUGID1-DRUGID&I; 
ARRAY PRESCRS(&I)   PRESCR1-PRESCR&I; 
ARRAY CONTSUBS(&I)  CONTSUB1-CONTSUB&I; 
ARRAY COMSTATS(&I)  COMSTAT1-COMSTAT&I; 
ARRAY &PASSAR; 
 
DO I=1 TO &I; 
   * RECODING MEDCODES TO MATCH DRUG FILE 2006; 
IF MEDS(I) IN ( '01025' '04535' '06885' '08960' '09185' '09360' '11060' '12395'  
    '14900' '15773' '17150' '19900' '24785' '24880' '28735'  
    '30870' '32685' '32680' '32690' '40215' '40335' '93240'  
    '93409' '94138' '96064' '96148' '97117') THEN MEDS(I)= '90000'; 
IF MEDS(I) NOT IN('90000' '88888') THEN DO; 
   IF MEDS(I) = '04055' THEN MEDS(I) = '96009'; 
   IF MEDS(I) = '04036' THEN MEDS(I) = '93371';  
   IF MEDS(I) = '04385' THEN MEDS(I) = '04370'; 
   IF MEDS(I) = '04410' THEN MEDS(I) = '00061'; 
   IF MEDS(I) = '08765' THEN MEDS(I) = '11615'; 
   IF MEDS(I) = '08780' THEN MEDS(I) = '96045'; 
   IF MEDS(I) = '14535' THEN MEDS(I) = '93195'; 
   IF MEDS(I) = '15043' THEN MEDS(I) = '40840'; 
   IF MEDS(I) = '15620' THEN MEDS(I) = '60735'; 
   IF MEDS(I) = '15945' THEN MEDS(I) = '15955'; 
   IF MEDS(I) = '19500' THEN MEDS(I) = '02109'; 
   IF MEDS(I) = '19820' THEN MEDS(I) = '19830'; 
   IF MEDS(I) = '20895' THEN MEDS(I) = '20890'; 
   IF MEDS(I) = '21213' THEN MEDS(I) = '98157'; 
   IF MEDS(I) = '23775' THEN MEDS(I) = '23765'; 
   IF MEDS(I) = '24408' THEN MEDS(I) = '24405'; 



   IF MEDS(I) = '25065' THEN MEDS(I) = '02161'; 
   IF MEDS(I) = '25470' THEN MEDS(I) = '42985'; 
   IF MEDS(I) = '28734' THEN MEDS(I) = '94051'; 
   IF MEDS(I) = '32385' THEN MEDS(I) = '32390'; 
   IF MEDS(I) = '32805' THEN MEDS(I) = '32800'; 
   IF MEDS(I) = '40170' THEN MEDS(I) = '02059'; 
   IF MEDS(I) = '40345' THEN MEDS(I) = '34560'; 
   IF MEDS(I) = '41440' THEN MEDS(I) = '27785'; 
   IF MEDS(I) = '42595' THEN MEDS(I) = '16210'; 
   IF MEDS(I) = '60335' THEN MEDS(I) = '06963'; 
   IF MEDS(I) = '60530' THEN MEDS(I) = '12620'; 
   IF MEDS(I) = '60690' THEN MEDS(I) = '14217'; 
   IF MEDS(I) = '60845' THEN MEDS(I) = '18020'; 
   IF MEDS(I) = '61040' THEN MEDS(I) = '21385'; 
   IF MEDS(I) = '61400' THEN MEDS(I) = '30240'; 
   IF MEDS(I) = '61450' THEN MEDS(I) = '30781'; 
   IF MEDS(I) = '89064' THEN MEDS(I) = '24885'; 
   IF MEDS(I) = '91011' THEN MEDS(I) = '40460'; 
   IF MEDS(I) = '92147' THEN MEDS(I) = '98110'; 
   IF MEDS(I) = '92169' THEN MEDS(I) = '93052'; 
   IF MEDS(I) = '92173' THEN MEDS(I) = '92175'; 
   IF MEDS(I) = '93031' THEN MEDS(I) = '04803'; 
   IF MEDS(I) = '93228' THEN MEDS(I) = '94127'; 
   IF MEDS(I) = '93343' THEN MEDS(I) = '16710'; 
   IF MEDS(I) = '93410' THEN MEDS(I) = '04020'; 
   IF MEDS(I) = '94083' THEN MEDS(I) = '15035'; 
   IF MEDS(I) = '94180' THEN MEDS(I) = '00324';  
   IF MEDS(I) = '96026' THEN MEDS(I) = '93336'; 
   IF MEDS(I) = '97102' THEN MEDS(I) = '02352'; 
 /*  ADDING ON NEW DRUG CHARACTERISTIC INFORMATION 
     BY USING FORMATS 
 */ 
 DRGID(I)  = PUT(MEDS(I), $DRUGID.); 
  IF DRGID(I) = 'NOMAT' THEN DO; 
   PUT 'NO MATCH FOR MED CODE,MISSING DRUG ID ' YEAR= MEDS(I)=; 
  END; 
 ELSE DO; 
 PRESCRS(I)  = PUT(INPUT(MEDS(I),5.),PRESCR.); 



 CONTSUBS(I) = PUT(INPUT(MEDS(I),5.),CONTSUB.); 
 COMSTATS(I) = PUT(INPUT(MEDS(I),5.),COMSTAT.); 
 CATEGORY    = PUT(MEDS(I),$RXCAT.); 
DO D=1 TO 16; 
 CATEG(I,D)=SUBSTR(CATEGORY,((3*D)-2),3); 
END; 
IF CATEG(I,1) = ' ' THEN DO; 
  /* WILL ONLY PRINT IF MATCH IN DATABASE 
   AND NO MULTUM CATEGORY ONE */ 
  PUT 'NO MATCH FOR MULTUM CATEGORY ONE' MEDS(I)=; 
 END; 
 END; 
END; 
END; 
DROP I D CATEGORY; 
 
%IF &VISYEAR = EDONLY  %THEN %DO; 
 DROP MED7 MED8; 
 /*RENAMING ADVERSE DRUG CHARACTERISTICS*/ 
 RENAME   
 DRUGID7=DRUGIDA1 
 PRESCR7=PRESCRA1 
 CONTSUB7=CONTSUBA1 
 COMSTAT7=COMSTATA1 
 RX7CAT1=RXA1CAT1 RX7CAT2=RXA1CAT2 RX7CAT3=RXA1CAT3 RX7CAT4=RXA1CAT4 
 RX7V1C1=RXA1V1C1 RX7V1C2=RXA1V1C2 RX7V1C3=RXA1V1C3 RX7V1C4=RXA1V1C4 
 RX7V2C1=RXA1V2C1 RX7V2C2=RXA1V2C2 RX7V2C3=RXA1V2C3 RX7V2C4=RXA1V2C4 
 RX7V3C1=RXA1V3C1 RX7V3C2=RXA1V3C2 RX7V3C3=RXA1V3C3 RX7V3C4=RXA1V3C4 
 DRUGID8=DRUGIDA2 
 PRESCR8=PRESCRA2 
 CONTSUB8=CONTSUBA2 
 COMSTAT8=COMSTATA2 
 RX8CAT1=RXA2CAT1 RX8CAT2=RXA2CAT2 RX8CAT3=RXA2CAT3 RX8CAT4=RXA2CAT4 
 RX8V1C1=RXA2V1C1 RX8V1C2=RXA2V1C2 RX8V1C3=RXA2V1C3 RX8V1C4=RXA2V1C4 
 RX8V2C1=RXA2V2C1 RX8V2C2=RXA2V2C2 RX8V2C3=RXA2V2C3 RX8V2C4=RXA2V2C4 
 RX8V3C1=RXA2V3C1 RX8V3C2=RXA2V3C2 RX8V3C3=RXA2V3C3 RX8V3C4=RXA2V3C4 
; 
%END; 



RUN; 
%MEND; 
%CREATEMED; 
 
/*CREATING PERMANENT OUTPUT DATASET*/ 
 
DATA LIBOUT.&VISOUT; 
SET VISNDC; 
RUN; 
 
PROC CONTENTS DATA=LIBOUT.NEWDRUGCHAR;  
RUN; 
 
 


